

Laskuvarjohyppy

Heittoliikkeeseen kuuluu keskeisinä tekijöinä maan vetovoima sekä ilmanvastuksen aiheuttama nopeudelle vastakkaissuuntainen voima. Näitä voimia sovelletaan myös mallinnettaessa laskuvarjolla hyppäämistä.

Tarkastellaan ensin vapaata putoamista. Putoajaan vaikuttava voima aiheuttaa kiihtyvyyden, joka saa aikaan nopeuden. Yleisessä muodossa voimalle pätee

$$F = m a,$$

missä m on tarkasteltavan kohteen massa ja a kiihtyvyys. Hyppääjään vaikuttavat voimat ovat painovoima $G = m g$, missä g on maan vetovoiman aiheuttama kiihtyvyys, sekä ilmanvastusvoima, jota arvioidaan lausekkeella

$$F = -b v^2 \hat{v} = -b v v.$$

Tässä b on ilmanvastuskerroin, v skalaarinen nopeus, $\mathbf{v}$ nopeus vektorina ja $\hat{v}$ nopeuden suuntainen yksikkövektori. Liikkeyhtälö saa tällöin muodon

$$m \frac{d^2 \mathbf{r}}{dt^2} = m \mathbf{g} - b \left| \frac{d \mathbf{r}}{dt} \right| \frac{d \mathbf{r}}{dt},$$

missä on merkitty $a = \frac{d^2 \mathbf{r}}{dt^2}$ ja $\mathbf{v} = \frac{d \mathbf{r}}{dt}$.

Tämä on heittoliikkeen differentiaaliyhtälö yleisessä muodossa.

Tehdään tarvittavat laskut laskuvarjohyppääjän lentoradan ja -nopeuden selvittämiseksi. Edellä esitetty ilmanvastusmalli on vain approksimaatio todelliselle ilmanvastukselle. On myös huomattava, että mikäli hyppääjä putoaa ilman varjoa, ilmanvastustekijä on aivan eri suuruinen kuin varjon kanssa.

Laskujen aluksi on syytä hävittää mahdollisista aiemmista laskuista jääneet muuttujat.

```
Remove["Global`*"]
```

Määritellään differentiaaliyhtälö normaaliryhmänä, jossa on erikseen x ja y -komponentit.

```
ryhma = {x'[t] == u[t], m u'[t] == -b Sqrt[u[t]^2 + v[t]^2] u[t],  
         y'[t] == v[t], m v'[t] == -m g - b Sqrt[u[t]^2 + v[t]^2] v[t]}
```

```
{x'[t] == u[t], m u'[t] == -b u[t] Sqrt[u[t]^2 + v[t]^2],  
 y'[t] == v[t], m v'[t] == -m g - b v[t] Sqrt[u[t]^2 + v[t]^2]}
```

Probleeman tuntemattomat funktiot ovat vaakasijaintia kuvaava $x(t)$ sekä korkeus $y(t)$ ja vastaavat nopeudet $u(t)$ ja $v(t)$.

```
tuntemattomat = {x[t], y[t], u[t], v[t]}
{x[t], y[t], u[t], v[t]}
```

Määritetään parametrit ja valitaan alkuehdot lentokoneesta hypättäessä. Oletetaan, että kyseessä on hitaasti lentävä koulutuslentokone.

```
m = 70; b = 30; g = 9.81;
alkuehto = {x[0] == 0, y[0] == 1000, u[0] == 20, v[0] == 0}
{x[0] == 0, y[0] == 1000, u[0] == 20, v[0] == 0}
```


Ratkaistaan differentiaaliyhtälöryhmä.

```
rtk = NDSolve[Flatten[{ryhma, alkuehto}], tuntemattomat, {t, 0, 250}]
{{x[t] → InterpolatingFunction[{{0., 250.}}, <>][t],
 y[t] → InterpolatingFunction[{{0., 250.}}, <>][t],
 u[t] → InterpolatingFunction[{{0., 250.}}, <>][t],
 v[t] → InterpolatingFunction[{{0., 250.}}, <>][t]}}
```

Muokataan tulos käyttökelpoiseen muotoon.

```
rata = tuntemattomat[[{1, 2}]] /. First[rtk]
{InterpolatingFunction[{{0., 250.}}, <>][t], InterpolatingFunction[{{0., 250.}}, <>][t]}
nopeus = tuntemattomat[[{3, 4}]] /. First[rtk]
{InterpolatingFunction[{{0., 250.}}, <>][t], InterpolatingFunction[{{0., 250.}}, <>][t]}
```

Piirretään kuva laskuvarjohyppäjän lentoradasta. Huomaa koordinaattiakselien eri skaalat!



```
FindRoot[rata[[2]] == 0, {t, 200}]
```


```
FindRoot::lstol :
```

```
The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal
but was unable to find a sufficient decrease in the merit function. You may need more
than MachinePrecision digits of working precision to meet these tolerances. More...
```

```
{t -> 209.502}
```

Lentorata ei muodostunut kovin pitkäksi ja laskeutuminen kilometrin korkeudelta tapahtui n. 3.5 minuutissa. Tarkastellaan vielä hyppääjän nopeutta.

```
Plot[Evaluate[nopeus], {t, 0, 250},
PlotStyle -> {Hue[0.3], Hue[0.6]}, PlotRange -> {{0, 5}, {-6, 20}}]
```



- Graphics -

```
nopeus /. Table[{t -> k}, {k, 1, 10}] // TableForm
```

1.39847	-4.27035
0.187609	-4.76782
0.0241775	-4.78395
0.00311125	-4.78434
0.000400353	-4.78435
0.0000515184	-4.78435
6.62905×10^{-6}	-4.78435
8.52915×10^{-7}	-4.78435
1.09537×10^{-7}	-4.78435
1.42179×10^{-8}	-4.78435

Vaakasuntainen nopeus lähes häviää ja putoamisnopeus asettuu vajaaseen 4.8 m/s nopeuteen jo kahden ensimmäisen sekunnin aikana. Hyppääjän putoamisvauhti maahan on siis noin 17 km/h, joka on suurenpuoleinen jaloilla vastaanotettavaksi.

Tarkastellaan seuraavaksi tilannetta, jossa hyppääjä ei avaa varjoaan välittömästi. Tehdään vastaava tarkastelu kuin edellä, paitsi että ilmanvastuskerroin b muuttuu hyppääjän avatessa varjon. Oletetaan, että ilman varjoa hyppäävän ihmisen maksiminopeus (vaikuttavien voimien summa on 0) eli ns. terminaalinopeus on runsaat 200 km/h eli n. 60 m/s. Tällöin ilmanvastuskerroin voidaan ratkaista yhtälöstä $F = mg - bv^2 = 0$, jolloin $b = mg / v^2$.

$$b_0 = 70 * 9.81 / 60^2$$


$$0.19075$$

Jotta ilmanvastuskertoimen arvo $b_0 = 0.19075$ voidaan varjon auetessa muuttaa arvoon $b = 30$, tarvitaan funktio, joka mallintaa muutoksen. Tällaiseksi sopii

$$f[x_] := 1 - \text{Exp}[-\text{Exp}[2x - 0.5]]$$

jonka kuvaaja näyttää seuraavalta:

`Plot[f[x], {x, -10, 10}, PlotRange -> {-0.5, 1.5}, PlotStyle -> Hue[.3]]`



- Graphics -

Määritetään putoamisyhtälöt uudestaan. Ilmanvastuskerroin määräytyy nyt funktiosta

$$bb[t_] := b_0 + (b - b_0) f[t]$$

ja uusi yhtälöryhmä on

$$\text{ryhma2} = \{x'[t] == u[t], m u'[t] == -bb[t - 20] \text{Sqrt}[u[t]^2 + v[t]^2] u[t], \\ y'[t] == v[t], m v'[t] == -mg - bb[t - 20] \text{Sqrt}[u[t]^2 + v[t]^2] v[t]\}$$

$$\{x'[t] == u[t], 70 u'[t] == (-0.19075 - 29.8093 (1 - e^{-e^{-0.5+2(-20+t)}})) u[t] \sqrt{u[t]^2 + v[t]^2}, \\ y'[t] == v[t], 70 v'[t] == -686.7 - (0.19075 + 29.8093 (1 - e^{-e^{-0.5+2(-20+t)}})) v[t] \sqrt{u[t]^2 + v[t]^2}\}$$

Varjon on oletettu aukeavan 20 sekuntia hyppämishetken jälkeen.

Ratkaistaan yhtälöryhmä.

`rtk2 = NDSolve[Flatten[{ryhma2, alkuehto}], tuntemattomat, {t, 0, 50}]`

```
{{x[t] -> InterpolatingFunction[{{0., 50.}}, <>][t],
 y[t] -> InterpolatingFunction[{{0., 50.}}, <>][t],
 u[t] -> InterpolatingFunction[{{0., 50.}}, <>][t],
 v[t] -> InterpolatingFunction[{{0., 50.}}, <>][t]}}
```

```


rata2 = tuntemattomat[{{1, 2}}] /. First[rtk2]
{InterpolatingFunction[{{0., 50.}}, <>][t], InterpolatingFunction[{{0., 50.}}, <>][t]}

nopeus2 = tuntemattomat[{{3, 4}}] /. First[rtk2]
{InterpolatingFunction[{{0., 50.}}, <>][t], InterpolatingFunction[{{0., 50.}}, <>][t]}

```

Piirretään kuvaaja hyppääjän lentoradasta, määritetään alastulohetki ja alastulonopeus:

```
ParametricPlot[rata2, {t, 0, 50}, PlotStyle -> Hue[.3], PlotRange -> {{0, 200}, {0, 1000}}]
```



- Graphics -

```
alas = FindRoot[rata2[[2]] == 0, {t, 50}]
```

```
{t -> 44.9826}
```

```
nopeus2 /. alas
```


```
{3.57963 × 10-14, -4.78435}
```

Tarkastellaan vielä putoamisnopeuksia kokonaistilanteen hahmottamiseksi. Putoamisnopeudet ajan funktiona ovat

```

Plot[Evaluate[nopeus2], {t, 0, 50}, PlotStyle -> {Hue[0.3], Hue[0.6]},
PlotRange -> {{0, 50}, {-60, 20}}, GridLines -> {{17, 20}, {}}]

```



- Graphics -


```
rata2 /. t -> 17
```

```
{163.553, 242.609}
```

Putoaminen tapahtuu varsin eri tavalla kuin avoimella varjolla. Vaikka varjon avautumishetkeksi asetettiin 20 s, avautuminen on alkanut askefunktion mukaisesti muuttamaan ilmanvastuskerrointa radikaalisti ja siten jarruttamaan putoamista jo aikaisemmin. Nopeus alkaa vähetä noin 250 metrin korkeudessa, kun $t = 17$ s.

Kiihtyvyys:

```
Plot[Evaluate[D[nopeus2[[2]], t]], {t, 0, 50},
      PlotRange -> All, PlotStyle -> Hue[0.6], GridLines -> {{19}, {}}]
```



- Graphics -

```
D[nopeus2[[2]], t] /. t -> 19
```

```
28.0379
```

Voimakkain jarrutuskiihtyvyys (n. $28 \frac{\text{m}}{\text{s}^2}$) tapahtuu kohdassa $t = 19$ s. Putoamisen kokonaisajasta avatun varjon kanssa kuluu yhä yli puolet, vaikka matka on noin viidennes.

Tehtävä

Milloin hyppääjän on viimeistään vedettävä varjon avausnarusta, kun oletetaan, että tämän on tapahduttava 3 sekuntia ennen askelfunktiossa käytettyä aikaa? Suurin mahdollinen nopeus maahan pudotessa ilman loukkaantumista on 10 m/s. Mikä on tällöin putoamisen kokonaisaika?

Linkit

[tykillä ampuminen](#) (tykki2.nb)
[normaaliryhmä](#)

JP & SKK 03.05.2001